

· #568

SOLPRO SUBROUTINE

MZ-14B

Table of Contents

1. Introduction
2. Errata/Change Log
3. LINKS TO RELEVANT INFORMATION IN THE ONLINE NSSDC
INFORMATION SYSTEM
4. Catalog Materials
 - a. Associated Documents
 - b. Core Catalog Materials

1. INTRODUCTION:

The documentation for this data set was originally on paper, kept in NSSDC's Data Set Catalogs (DSCs). The paper documentation in the Data Set Catalogs have been made into digital images, and then collected into a single PDF file for each Data Set Catalog. The inventory information in these DSCs is current as of July 1, 2004. This inventory information is now no longer maintained in the DSCs, but is now managed in the inventory part of the NSSDC information system. The information existing in the DSCs is now not needed for locating the data files, but we did not remove that inventory information.

The offline tape datasets have now been migrated from the original magnetic tape to Archival Information Packages (AIP's).

A prior restoration may have been done on data sets, if a requestor of this data set has questions; they should send an inquiry to the request office to see if additional information exists.

2. ERRATA/CHANGE LOG:

NOTE: Changes are made in a text box, and will show up that way when displayed on screen with a PDF reader.

When printing, special settings may be required to make the text box appear on the printed output.

Version	Date	Person	Page	Description of Change
01				
02				

3 LINKS TO RELEVANT INFORMATION IN THE ONLINE NSSDC
INFORMATION SYSTEM:

<http://nssdc.gsfc.nasa.gov/nmc/>

[NOTE: This link will take you to the main page of the NSSDC Master Catalog. There you will be able to perform searches to find additional information]

4. CATALOG MATERIALS:

- a. Associated Documents To find associated documents you will need to know the document ID number and then click here.
<http://nssdcftp.gsfc.nasa.gov/miscellaneous/documents/>

- b. Core Catalog Materials

SOLPRO SUBROUTINE

MZ-14B

This data set catalog consists of one 9-track, 1600 BPI, ASCII magnetic tape with one file of data. The tape was created on the MODCOMP IV computer. The D and C numbers are listed below.


D#

D-40374

C#

C-23653

B 23274-000A



SOLPRO: A COMPUTER CODE TO CALCULATE PROBABILISTIC ENERGETIC SOLAR PROTON FLUENCES

E. G. STASSINOPOULOS

APRIL 1975



NATIONAL SPACE SCIENCE DATA CENTER

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION • GODDARD SPACE FLIGHT CENTER, GREENBELT, MD.

SOLPRO: A Computer Code to Calculate
Probabilistic Energetic Solar Proton Fluences

E. G. Stassinopoulos

NASA-Goddard Space Flight Center
Sciences Directorate
National Space Science Data Center

April 1975

Greenbelt, Maryland 20771

ABSTRACT

A code has been developed for the calculation of interplanetary solar proton fluences at 1 A. U. for the active years 1977-1983. The fluences are presented as functions of mission duration τ , energy threshold E , and confidence level Q . For a given combination of τ and Q , the routine determines whether ordinary or anomalously large events are to be considered, and in the latter case, the number of anomalously large events that are predicted by probabilistic theory for the specified mission duration. The code is described in detail. A listing and sample calculations are attached.

CONTENTS

	<u>Page</u>
1. Introduction	1
2. Method and Procedure	4
3. Results and Discussion	8
4. Application and Use	9

Acknowledgements

References

Attachments: Tables and Figures

Program Listing and Sample Calculations

List of Tables

Table

1	Polynomial Expansion Coefficients A_{ij} for Number of AL
2	Number of AL Events Predicted by SOLPRO
3	OR-Event Fluences Predicted by SOLPRO
4	Polynomial Expansion Coefficients B_{ij} for OR-Event Flue
5	AL-Event Spectrum and OR-Event Distribution

List of Figures

Figure

1	Comparison of OR-Event Fluences $J(E > 30 \text{ Mev})$
2	Comparison of Number of AL Events
3	SOLPRO Flow Diagram

SOLPRO: A Code to Calculate Energetic Solar Proton Fluences

Introduction

The code was originally developed as a supplement to UNIFLUX, a unified orbital flux integration system (Stassinopoulos and Gregory, 1975), from an initial solar proton model (Stassinopoulos and King, 1974) designed exclusively for the evaluation of flux levels to be encountered during the active phase of the next solar cycle (1977-1983) by earth-orbiting spacecraft on missions involving partial magnetospheric shielding.

This first empirical model was characteristically simple and easy to use. It was derived entirely from experimental satellite measurements made during the active years of solar cycle 20; it included only crude confidence level estimates, that is, the probability of actual cycle 21 fluxes exceeding the predicted intensities.

A more probabilistic analysis of solar cycle 20 fluxes was later given by King (1974). This formal statistical treatment was subsequently used by King and Stassinopoulos (1975) in their computation of ratios of solar to trapped proton fluences for circular-orbit geocentric space missions to be flown during the active years of the next solar cycle (1977-1983). In this more sophisticated model the probability of exceeding mission fluence levels is given as a function of mission duration and energy.

The new SOLPRO code incorporates all the important features of the King (1974) analysis, which includes the distinction between "ordinary" (OR) and "anomalously large" (AL) events, and the probability of occurrence of the latter. Be it noted, that only one anomalously large event

was observed during solar cycle 20: the event of August, 1972; according to King (1974), the proton fluences of that one event alone constituted about 69% ($E > 10$ MeV) and 84% ($E > 60$ MeV) of the entire cycle fluence. Consequently, whenever at least one AL event is predicted, the OR-event-contribution is negligible.

The referenced solar proton models are based on the most continuous set of satellite data available to date, covering the period of enhanced solar activity during the 20th solar cycle (1964-1975) through the 1972 AL event, after which no significant events occurred. In both models, all fluxes were taken to be isotropic. As noted in Stassinopoulos and King (1974), on an event-integrated basis, departures from isotropy are typically only a few percent or less. This applies to the interplanetary medium and to much of the magnetosphere, but may not apply at low altitudes where anisotropy should result from atmospheric loss mechanisms.

When using the SOLPRO routine, two important items should be kept in mind. First, there is no assurance that the overall flux levels observed during the 20th solar cycle will also occur during the 21st cycle. However, as indicated in King (1974), a comparison of the annual mean sunspot numbers of past solar cycles suggests that cycle 21 will most likely be similar to the very ordinary cycle 20 than to the very extraordinary cycle 19. Second, there is no reliable way of predicting the distribution of individual solar events in time, in flux level, and in spectra through the 21st cycle. That pertains to ordinary as well as to anomalously large events. Therefore, the results obtained from the routine, although the best predictions available of solar proton fluences expected to occur during 1977-1983, should be considered guideline approximations only.

In the code, the solar proton fluences are expressed as functions of the parameters "confidence level", Q^* (in percent), "mission duration", τ (in months), and energy, E (in MeV). Within the limits established by the routine the user is free to select his own values of Q and τ for processing, but E is not an input parameter and the calculations are performed at ten fixed energy thresholds.

For missions whose trajectories will involve significant time away from a heliocentric distance of 1 A.U. (astronomical unit), a helioradial dependence of event fluences must be allowed for, as discussed by King (1974).

Similarly, for missions whose trajectories will involve partial magnetospheric shielding, the fractional exposure factor has to be determined, as discussed by Stassinopoulos and King (1974). An independent code performing these shielding calculations has been developed (Stassinopoulos, 1975) and is available from the National Space Science Data Center, Greenbelt, Maryland.

A detailed description of the solar proton routine is given in the next section on "Method and Procedure", with an analysis of the method employed and a review of the organization and structure of the code. In section "Application and Use" the routine is briefly discussed, its operation is described, and the arguments of its transfer vector are presented. Code listing and sample outputs are given in the attachments.

* Q is the degree of confidence a mission planner wishes to have that the *actually encountered* solar proton fluence will not exceed the *predicted* fluence, as given by the statistical solar proton model used.

Method and Procedure

SOLPRO is a short, compact routine which evaluates omnidirectional, unattenuated, interplanetary energetic solar proton fluences in terms of three variables: Q , τ , and E . The first two are input parameters, the last is internally programmed. As defined elsewhere in this report, Q denotes the degree of confidence one wishes to assign to the results, namely that for the specified mission duration τ the calculated fluences at ten preset energy values ranging from 10 to 100 MeV in increments of 10 MeV, are the smallest values which will not be exceeded by actually encountered intensities.

In order to hold the size and complexity of the code to a minimum, the desirable elements of the King (1974) analysis were empirically formulated by convenient approximation techniques, which are able to reproduce the statistically treated King results with relatively high accuracy. Specifically, the maximum error introduced into the statistics by this method is less than 50%. This error is indeed insignificant in view of the large intrinsic statistical uncertainties of the models themselves and the much greater uncertainties involved in predicting cycle 21 fluxes from cycle 20 data.

The most important feature of the code is the distinction between OR and AL events. This capability was developed relying entirely on King's probabilistic treatment and using the data given therein (King, 1974; Figure 5). For simplicity let:

$$n = f(Q, \tau) \quad (1)$$

where n is the number of AL events expected to occur with a confidence level Q and for a mission duration τ . It is then convenient to effect an apparent separation of variables by evaluating n independently for every value of Q within the regime of interest, and by expressing n in terms of τ alone. Thus:

$$n_i = c + \sum_{j=0}^6 a_{ij}(Q) \tau^j \quad (2)$$

where the index i relates n to Q through:

$$i = 100 - Q \quad (3)$$

and where the polynomial expansion coefficients a_{ij} are determined by curve fitting. The constant, $c=1.001$, is a dimensionless range adjustment term.

Table 1 contains the numerical values of the a_{ij} coefficients.

The code calculates n at the very beginning from equation 2 (see Table 2 for obtained values of n) and, depending on whether AL events are expected to occur or not, it branches to its appropriate section, where it computes the corresponding AL or OR solar proton fluences for the specified Q and τ .

For AL-event conditions, that is $n > 0$, the fluences are obtained from the expression:

$$F_{AL}(>E; Q, \tau) = S_{AL}(>E) * n(Q, \tau) \quad (4)$$

where the function:

$$S_{AL}(>E) = 7.9 \times 10^9 \exp \{ (30. - E) / 26.5 \} \quad (5)$$

is King's (1974) exponential-in-energy representation of the integral flux and spectrum of the August, 1972, anomalously large event, for energy thresholds between 10 and 200 MeV.

In case OR events only are predicted ($n=0$), the respective fluences are obtained from:

$$F_{OR}(>E; Q, \tau) = S_{OR}(>E) * J(Q, \tau) \quad (6)$$

where now King's (1974) expression:

$$S_{OR}(>E) = \exp \{ .0158(30.-E) \} \quad (7)$$

for the integral energy spectrum of ordinary events is used, reliable over the range $10 \leq E(\text{MeV}) \leq 100$.

The function $J(Q, \tau)$ gives the ordinary fluence of protons above 30 MeV in terms of the two input variables (see Table 3 for obtained values of J). It was developed along similar lines as equation 1, with data from the same source (King, 1974; Figure 8).

Again, through the independent evaluation of J , (but this time for every τ within the relevant interval of 1-9 months for these events) the same apparent separation of variables is effected, resulting in:

$$J_k = d \sum_{m=0}^4 b_{km}(\tau) Q^m \quad (8)$$

where the index k relates J to τ through:

$$k = \text{integer}(\tau) \quad (9)$$

and where the coefficients, contained in Table 4, are also determined

by curve fitting. The constant, $d = 10^7$, is a scaling factor.

The code is structurally complete and self-contained. All expansion coefficients for the statistical treatment are stored in data statements at the beginning of the routine, while the expressions for the AL and OR models are included in the respective sections of the code.

To guard against accidental misuse and so as to insure that calculations are only performed within the valid ranges of the independent variables Q and τ , SOLPRO tests the values of these input parameters in its first executable statement. If the parameter(s) exceeds the program limits, it prints a warning message before returning control to the calling routine.

Finally, the code bypasses ordinary event fluence calculations, although it branches to the appropriate section, whenever $\tau=1$ and Q has values ranging from 80 to 83. In that domain the flux gradients are very high, the intensities decrease very rapidly towards $Q = 80$ and the galactic cosmic ray component becomes predominant (see discussion by King, 1974). For all practical purposes the OR-event solar protons are negligible for $Q < 80$ and $\tau > 1$.

Results and Discussion

Sample calculations performed with the code indicate good agreement with probabilistic predictions. That is, the results represent a reasonable approximation to these data, considering the large intrinsic limitations and uncertainties inherent in this type of estimation process.

Thus, as shown in Figure 1, the ordinary event fluence J of protons with energies $E > 30$ MeV, computed by the routine for the applicable Q and τ ranges, compares well with the original data in King (1974; Figure 8); the solid curves indicate the former, the dotted curves the latter. Apparently, the curves overlap for most Q and τ values. In the few places where they do not agree, the differences are truly insignificant; for example, the average error for all $\tau > 1$ is about 3%, while the maximum error, which occurs at $\tau = 1$ for $Q = 85$ is less than 50%.

Similarly, the number of anomalously large events n given by the code agrees favorably with the model (King, 1974; Figure 5). This is shown in Figure 2, where the dots indicate model data and the triangles SOLPRO predictions. Since the transition from one n value to another is probabilistically (and not deterministically) defined, the differences between the King model and the SOLPRO approximation are inconsequential.

Finally, the AL event exponential-in-energy function (Equation 5) and the ordinary-event integral-energy spectrum (Equation 7) are identical to the formulae given by King (1974). Their respective values are listed in Table 5.

Application and Use

Subroutine SOLPRO is written in standard FORTRAN-IV computer language and card decks are available in either the 029 model IBM keypunch format (EBCDIC) or the 026 keypunch format (BCD), for use with FORTRAN-compatible compilers.

The variables in the calling sequence are:

SOLPRO(TAU,IQ,F)

The first two arguments are input data:

TAU : mission duration in months, with a valid range
from 1. to 72. (REAL*4)

IQ : percent confidence level that calculated fluence
will not be exceeded, with a valid range from
80 to 99 (INTEGER*4)

The last argument is output data:

F(N) : solar proton spectrum for energies $E > 10 \cdot N$ MeV,
where index N goes from 1 to 10.

All arguments except IQ are single precision floating point variables. Besides their range limits, there are no other restrictions or limitations on the values of the input arguments.

There are no READ or other input statements in SOLPRO. A WRITE statement is executed each time an input variable exceeds the stipulated limits. In addition, a WRITE statement is executed whenever for a given combination of τ and Q no significant solar proton fluences are to be expected.

Figure 3 shows a flow diagram of the program. A complete listing is given in the attachment. The cards of the deck are appropriately labeled in columns 73-80 as SOLPRxxx where the three last columns (xxx) contain the sequential numbering, which is incremented by 10.

Acknowledgements

The author wishes to thank Drs. J. I. Vette and J. H. King
for many helpful suggestions and comments.

References

King, J. H., "Solar Proton Fluences for 1977-1983 Space Missions", J. Spacecraft and Rockets, Vol. 11, No. 6, pp.401-408, June, 1974.

King, J. H., and E. G. Stassinopoulos, "Energetic Solar Protons vs. Terrestrially Trapped Proton Fluxes for the Active Years 1977-1983", J. Spacecraft and Rockets, Vol. 12, No. 2, pp. 122-124, February, 1975.

Stassinopoulos, E. G., and J. H. King, "Empirical Solar Proton Model for Orbiting Spacecraft Application", IEEE Transactions on Aerospace and Electronic Systems, Vol. 10, No. 4, July 4, 1974.

Stassinopoulos, E. G., and C. Z. Gregory, "UNIFLUX: A Unified Orbital Flux Integration and Analysis System", to be published in 1975.

Stassinopoulos, E. G., "GMSHLD: A Computer Code to Calculate Cosmic Ray Geomagnetic Shielding Effects for Earth Orbiting Satellites", to be published in 1975.

Table 1


NUMBER OF ANOMALOUSLY LARGE EVENTS

Polynomial Expansion Coefficients

Q (%)	a_0	a_1	a_2	a_3	a_4	a_5	a_6
99	-.1571	.2707	-.1269 ⁻¹	.4428 ⁻³	-.8185 ⁻⁵	.7754 ⁻⁷	-.2939 ⁻⁹
98	-.1870	.1951	-.6559 ⁻²	.1990 ⁻³	-.3618 ⁻⁵	.3740 ⁻⁷	-.1599 ⁻⁹
97	-.2007	.1497	-.3179 ⁻²	.5730 ⁻⁴	-.4664 ⁻⁶	.1764 ⁻⁸	0
96	-.1882	.1228	-.1936 ⁻²	.2660 ⁻⁴	-.1022 ⁻⁶	0	0
95	-.2214	.1149	-.1871 ⁻²	.2695 ⁻⁴	-.1116 ⁻⁶	0	0
94	-.2470	.1062	-.1658 ⁻²	.2367 ⁻⁴	-.9465 ⁻⁷	0	0
93	-.2509	.8710 ⁻¹	-.8300 ⁻³	.8438 ⁻⁵	0	0	0
92	-.2923	.8932 ⁻¹	-.1023 ⁻²	.1029 ⁻⁴	0	0	0
91	-.3222	.8648 ⁻¹	-.9992 ⁻³	.9935 ⁻⁵	0	0	0
90	-.3518	.8417 ⁻¹	-.1000 ⁻²	.9956 ⁻⁵	0	0	0
89	-.3698	.7951 ⁻¹	-.8983 ⁻³	.8940 ⁻⁵	0	0	0
88	-.2771	.5473 ⁻¹	-.1543 ⁻⁴	0	0	0	0
87	-.2818	.5072 ⁻¹	.2511 ⁻⁴	0	0	0	0
86	-.2845	.4717 ⁻¹	.5664 ⁻⁴	0	0	0	0
85	-.2947	.4405 ⁻¹	.8507 ⁻⁴	0	0	0	0
84	-.2923	.4111 ⁻¹	.1106 ⁻³	0	0	0	0
83	-.2981	.3853 ⁻¹	.1312 ⁻³	0	0	0	0
82	-.3002	.3585 ⁻¹	.1529 ⁻³	0	0	0	0
81	-.3001	.3312 ⁻¹	.1781 ⁻³	0	0	0	0
80	-.3141	.3248 ⁻¹	.1654 ⁻³	0	0	0	0

Table 2

Number of Anomalous Large Events Predicted by SOLPRO



*NO CALCULATIONS PERFORMED

Table 3

OR-Event Fluences ($E > 30$ MeV) given by SOLPRO Routine

Q	1	2	3	4	5	6	7	8	9
99	-----	-----	-----	-----	-----	-----	-----	-----	-----
98	-----	-----	-----	-----	-----	-----	-----	-----	-----
97	5.07×10^8	-----	-----	-----	-----	-----	-----	-----	-----
96	3.42×10^8	-----	-----	-----	-----	-----	-----	-----	-----
95	2.34×10^8	-----	-----	-----	-----	-----	-----	-----	-----
94	1.67×10^8	4.83×10^8	-----	-----	-----	-----	-----	-----	-----
93	1.30×10^8	3.91×10^8	-----	-----	-----	-----	-----	-----	-----
92	1.09×10^8	3.23×10^8	6.46×10^8	-----	-----	-----	-----	-----	-----
91	9.78×10^7	2.74×10^8	5.09×10^8	-----	-----	-----	-----	-----	-----
90	8.85×10^7	2.39×10^8	4.15×10^8	7.82×10^8	-----	-----	-----	-----	-----
89	7.72×10^7	2.12×10^8	3.53×10^8	6.02×10^8	-----	-----	-----	-----	-----
88	6.20×10^7	1.91×10^8	3.14×10^8	4.84×10^8	9.16×10^8	-----	-----	-----	-----
87	4.34×10^7	1.73×10^8	2.88×10^8	4.12×10^8	6.76×10^8	-----	-----	-----	-----
86	2.40×10^7	1.56×10^8	2.70×10^8	3.68×10^8	5.44×10^8	-----	-----	-----	-----
85	8.86×10^6	1.38×10^8	2.53×10^8	3.42×10^8	4.74×10^8	7.52×10^8	-----	-----	-----
84	5.23×10^6	1.19×10^8	2.36×10^8	3.23×10^8	4.32×10^8	6.25×10^8	-----	-----	-----
83	*	1.00×10^8	2.17×10^8	3.04×10^8	3.98×10^8	5.39×10^8	8.06×10^8	-----	-----
82	*	8.16×10^7	1.95×10^8	2.83×10^8	3.60×10^8	4.80×10^8	6.94×10^8	1.01×10^9	-----
81	*	6.56×10^7	1.74×10^8	2.58×10^8	3.22×10^8	4.38×10^8	6.08×10^8	8.38×10^8	-----
80	*	5.48×10^7	1.56×10^8	2.32×10^8	2.98×10^8	4.04×10^8	5.41×10^8	7.34×10^8	1.01×10^9

* No Calculations performed; see comments page 7.

Table 4

ORDINARY EVENT FLUENCE

Polynomial Expansion Coefficients

τ (mths)	b_0	b_1	b_2	b_3	b_4
1	$.154047 \times 10^3$	$-.522258 \times 10^4$	$.714275 \times 10^5$	$-.432747 \times 10^6$	$.955315 \times 10^6$
2	$.198004 \times 10^3$	$-.448788 \times 10^4$	$.438148 \times 10^5$	$-.196046 \times 10^6$	$.325520 \times 10^6$
3	$.529120 \times 10^3$	$-.122227 \times 10^5$	$.112869 \times 10^6$	$-.465084 \times 10^6$	$.710572 \times 10^6$
4	$.121141 \times 10^4$	$-.266412 \times 10^5$	$.226778 \times 10^6$	$-.857230 \times 10^6$	$.120444 \times 10^7$
5	$.452062 \times 10^4$	$-.103248 \times 10^6$	$.896085 \times 10^6$	$-.346028 \times 10^7$	$.499852 \times 10^7$
6	$.272028 \times 10^4$	$-.499088 \times 10^5$	$.353050 \times 10^6$	$-.111929 \times 10^7$	$.133386 \times 10^7$
7	$.275595 \times 10^4$	$-.469718 \times 10^5$	$.314729 \times 10^6$	$-.960383 \times 10^6$	$.111650 \times 10^7$
8	$.570997 \times 10^4$	$-.799689 \times 10^5$	$.381074 \times 10^6$	$-.610714 \times 10^6$	0
9	$.101000 \times 10^3$	0	0	0	0

Table 5

Integral Flux & Spectrum
of the August 1972 AL Event
(Equation 5)

Distribution of
OR-Event Integral Spectrum
(Equation 7)

<u>Energy</u> (MeV)	<u>Solar Protons (>E)</u> (#/cm ² /AL-event)	<u>Energy</u> (MeV)	<u>Spectral Function</u> (normalized to 30 MeV)
10	1.680×10^{10}	10	1.37163
20	1.152×10^{10}	20	1.17117
30	7.900×10^9	30	1.00000
40	5.417×10^9	40	0.85385
50	3.714×10^9	50	0.72906
60	2.547×10^9	60	0.62251
70	1.746×10^9	70	0.53153
80	1.197×10^9	80	0.45384
90	8.210×10^8	90	0.38752
100	5.629×10^8	100	0.33088




Figure 1. OR-Event Fluences ($E > 30 \text{ MeV}$): Comparison of Predictions




Figure 2. Number of Anomalously Large Events: Comparison of Predictions




Figure 3. Flow Diagram for SOLPRO

Sample Program Listing

```

      SUBROUTINE SOLPRO(TAU,IQ,F)
C *** INTERPLANETARY SOLAR PROTON FLUX AT 1 AU (FROM E>10 TO E>100 MEV)
C *** SINGLE PRECISION DECK IN STANDARD FORTRAN IV FOR IBM 360 MACHINES
C *** (EBCDIC, 029 PUNCH) OR OTHER COMPATIBLE SYSTEMS.
C *** PROGRAM DESIGNED AND TESTED BY E.G. STASSINOPOULOS, CODE 601,
C *** NASA GODDARD SPACE FLIGHT CENTER, GREENBELT, MARYLAND 20771 .
C *****
C **** INPUT: TAU      MISSION DURATION IN MONTHS (REAL*4)
C ****          IQ      CONFIDENCE LEVEL THAT CALCULATED FLUENCE F(N)
C ****                  WILL NOT BE EXCEEDED (INTEGER*4)
C **** OUTPUT: F(N)     SPECTRUM OF INTEGRAL SOLAR PROTON FLUENCE FOR
C ****                  ENERGIES E>10*N (1<N<10)
      REAL NALE,NALECF(7,20)/-.1571,.2707,-.1269E-1,.4428E-3,-.8185E-5,
      $.7754E-7,-.2939E-9,-.1870,.1951,-.6559E-2,.1990E-3,-.3618E-5,
      $.3740E-7,-.1599E-9,-.2007,.1497,-.3179E-2,.5730E-4,-.4664E-6,
      $.1764E-8,0,-.1882,.1228,-.1936E-2,.2660E-4,-.1022E-6,2*0.,
      $-.2214,.1149,-.1871E-2,.2695E-4,-.1116E-6,2*0.,-.2470,.1062,
      $-.1658E-2,.2367E-4,-.9465E-7,2*0.,-.2509,.8710E-1,-.8300E-3,
      $.8438E-5,3*0.,-.2923,.8932E-1,-.1023E-2,.1029E-4,3*0.,-.3222,
      $.8648E-1,-.9992E-3,.9935E-5,3*0.,-.3518,.8417E-1,-.1000E-2,
      $.9956E-5,3*0.,-.3698,.7951E-1,-.8983E-3,.8940E-5,3*0.,-.2771,
      $.5473E-1,-.1543E-4,4*0.,-.2818,.5072E-1,.2511E-4,4*0.,-.2845,
      $.4717E-1,.5664E-4,4*0.,-.2947,.4405E-1,.8507E-4,4*0.,-.2923,
      $.4111E-1,.1106E-3,4*0.,-.2981,.3853E-1,.1312E-3,4*0.,-.3002,
      $.3585E-1,.1529E-3,4*0.,-.3001,.3312E-1,.1781E-3,4*0.,-.3141,
      $.3248E-1,.1654E-3,4*0./,F(10),G(10)
      REAL ORFLXC(5,9)/.154047E3,-.522258E4,.714275E5,-.432747E6,.955315
      $E6,.198004E3,-.448788E4,.438148E5,-.196046E6,.32552E6,.529120E3,
      $-.122227E5,.112869E6,-.465084E6,.710572E6,.121141E4,-.266412E5,
      $.226778E6,-.85728E6,.120444E7,.452062E4,-.103248E6,.896085E6,
      $-.346028E7,.499852E7,.272028E4,-.499088E5,.35305E6,-.111929E7,
      $.133386E7,.275597E4,-.469718E5,.314729E6,-.960383E6,.11165E7,
      $.570997E4,-.799689E5,.381074E6,-.610714E6,0.,.101E3,4*0./
      INTEGER INDEX(20)/2*7,6,3*5,5*4,9*3/
      1 FORMAT(' TAU=',F4.0,' IQ=',I3,3X,'PARAMETER(S) EXCEED PROGRAM LIMISOLPR350
      $TS')
      2 FORMAT(2X,'FOR THE COMBINATION OF TAU AND IQ GIVEN, NO SIGNIFICANTSOLPR370
      $ SOLAR PROTON FLUXES ARE TO BE EXPECTED. TAU=',F6.2,' IQ=',I2)
      IF(TAU.GT.72..OR.IQ.LT.80)GO TO 500
      IP=100-IQ
      M=INDEX(IP)
      NALE=0.
      DO 300 J=1,M
      300 NALE=NALE+NALECF(J,IP)*TAU**(J-1)
      INALE=NALE+1.0001
      IF(INALE.GT.0) GO TO 400
C *** CALCULATIONS FOR OR-EVENT CONDITIONS
      IT=TAU
      IF(IT.EQ.1.AND.IP.GT.16) GO TO 700
      P=FLOAT(IP)/100.
      OF=0.
      DO 100 J=1,5
      100 OF=OF+ORFLXC(J,IT)* P**(J-1)*1.E7
      E=10.
      DO 200 N=1,10
      G(N)=EXP(.0158*(30.-E))
      F(N)=OF*G(N)
      200 E=E+10.
      GO TO 800
C *** CALCULATIONS FOR AL-EVENT CONDITIONS
      400 E=10.
      DO 600 N=1,10
      F(N)=7.9E9*EXP((30.-E)/26.5)*INALE
      600 E=E+10.
      GO TO 800
      700 WRITE(6,2) TAU,IQ
      GO TO 800
      500 WRITE (6,1) TAU,IQ
      800 RETURN
      END

```

Sample Calculations: Solar Proton Fluences $F(>E; Q, \tau)$

[illegible][illegible][illegible][illegible][illegible][illegible]

Solpro
File 1 - Subroutine

```
1 SUBROUTINE SOLPRO%TAU,IQ,F,INALE< SOLPR010
2 C *** MODIFIED 9/77 TO RETURN INALE% OF AL EVENTS< TO CALLING PROGRAM SOLPR020
3 C *** INTERPLANETARY SOLAR PROTON FLUX AT 1 AU %FROM E=10 TO E=100 MEV< SOLPR030
4 C *** SINGLE PRECISION DECK IN STANDARD FORTRAN IV FOR IBM 360 MACHINES SOLPR040
5 C *** %EBCDIC, 029 PUNCH< OR OTHER COMPATIBLE SYSTEMS. SOLPR050
6 C *** PROGRAM DESIGNED AND TESTED BY E.G. STASSINOPOULOS, CODE 601, SOLPR060
7 C *** NASA GODDARD SPACE FLIGHT CENTER, GREENBELT, MARYLAND 20771 . SOLPR070
8 C ***** SOLPR080
9 C ***** INPUT' TAU MISSION DURATION IN MONTHS %REAL*4< SOLPR090
10 C ***** IQ CONFIDENCE LEVEL THAT CALCULATED FLUENCE F%N< SOLPR100
11 C ***** WILL NOT BE EXCEEDED %INTEGER*4< SOLPR110
12 C ***** OUTPUT' F%N< SPECTRUM OF INTEGRAL SOLAR PROTON FLUENCE FOR SOLPR120
13 C ***** ENERGIES E=10*N %1#*N#+10< SOLPR130
14 C ***** INALE OF AL EVENTS FOR GIVEN TAU AND Q SOLPR140
15 REAL NALE,NALECF%7,20</-.1571,.2707,-.1269E-1,.4428E-3,-.8185E-5, SOLPR150
16 $.7754E-7,-.2939E-9,-.1870,.1951,-.6559E-2,.1990E-3,-.3618E-5, SOLPR160
17 $.3740E-7,-.1599E-9,-.2007,.1497,-.3179E-2,.5730E-4,-.4664E-6, SOLPR170
18 $.1764E-8,0,-.1882,.1228,-.1936E-2,.2660E-4,-.1022E-6,2*0., SOLPR180
19 $-.2214,.1149,-.1871E-2,.2695E-4,-.1116E-6,2*0.,-.2470,.1062, SOLPR190
20 $-.1658E-2,.2367E-4,-.9465E-7,2*0.,-.2509,.8710E-1,-.8300E-3, SOLPR200
21 $.8438E-5,3*0.,-.2923,.8932E-1,-.1023E-2,.1029E-4,3*0.,-.3222, SOLPR210
22 $.8648E-1,-.9992E-3,.9935E-5,3*0.,-.3518,.8417E-1,-.1000E-2, SOLPR220
23 $.9956E-5,3*0.,-.3698,.7951E-1,-.8983E-3,.8940E-5,3*0.,-.2771, SOLPR230
24 $.5473E-1,-.1543E-4,4*0.,-.2818,.5072E-1,.2511E-4,4*0.,-.2845, SOLPR240
25 $.4717E-1,.5664E-4,4*0.,-.2947,.4405E-1,.8507E-4,4*0.,-.2923, SOLPR250
26 $.4111E-1,.1106E-3,4*0.,-.2981,.3853E-1,.1312E-3,4*0.,-.3002, SOLPR260
27 $.3585E-1,.1529E-3,4*0.,-.3001,.3312E-1,.1781E-3,4*0.,-.3141, SOLPR270
28 $.3248E-1,.1654E-3,4*0./,F%10<,G%10< SOLPR280
29 REAL ORFLXC%5,9</.154047E3,-.522258E4,.714275E5,-.432747E6,.955315E6, SOLPR290
30 $E6,.198004E3,-.448788E4,.438148E5,-.196046E6,.32552E6,.529120E3, SOLPR300
31 $-.122227E5,.112869E6,-.465084E6,.710572E6,.121141E4,-.266412E5, SOLPR310
32 $.226778E6,-.85728E6,.120444E7,.452062E4,-.103248E6,.896085E6, SOLPR320
33 $-.346028E7,.499852E7,.272028E4,-.499088E5,.35305E6,-.111929E7, SOLPR330
34 $.133386E7,.275597E4,-.469718E5,.314729E6,-.960383E6,.11165E7, SOLPR340
35 $.570997E4,-.799689E5,.381074E6,-.610714E6,0.,.101E3,4*0./ SOLPR350
36 INTEGER INDEX%20</2*7,6,3*5,5*4,9*3/ SOLPR360
37 1 FORMAT%@ TAU#@,F4.0,@ IQ#@,I3,3X,@PARAMETER%$< EXCEED PROGRAM LIMISOLPR370
38 $TS@< SOLPR380
39 2 FORMAT%2X,@FOR THE COMBINATION OF TAU AND IQ GIVEN, NO SIGNIFICANTSOLPR390
40 $ SOLAR PROTON FLUXES ARE TO BE EXPECTED. TAU#@,F6.2,@ IQ#@,I2< SOLPR400
41 IF%TAU.GT.72..OR.IQ.LT.80<GO TO 500 SOLPR410
42 IP#100-IQ SOLPR420
43 MHINDEX%IP< SOLPR430
44 VALE#0. SOLPR440
45 DO 300 J#1,M SOLPR450
46 300 NALE#NALE&NALECF%J,IP<*TAU**%J-1< SOLPR460
47 INALE#NALE&1.0001 SOLPR470
48 IF%INALE.GT.0< GO TO 400 SOLPR480
49 C *** CALCULATIONS FOR OR-EVENT CONDITIONS SOLPR490
50 IT#TAU SOLPR500
51 IF%IT.EQ.1.AND.IP.GT.16< GO TO 700 SOLPR510
52 P#FLOAT%IP</100. SOLPR520
53 OF#0. SOLPR530
54 DO 100 J#1,5 SOLPR540
55 100 OF#OF&ORFLXC%J,IT<* P**%J-1<*1.E7 SOLPR550
56 E#10. SOLPR560
57 DO 200 N#1,10 SOLPR570
58 G%N<#EXP%.0158*%30.-E<< SOLPR580
```

59	F%N<#OF*G%N<	SOLPR590
60	200 E#E&10.	SOLPR600
61	GO TO 800	SOLPR610
62	C *** CALCULATIONS FOR AL-EVENT CONDITIONS	SOLPR620
63	400 E#10.	SOLPR630
64	DO 600 N#1,10	SOLPR640
65	F%N<#7.9E9*EXP%%30.-E</26.5<*INALE	SOLPR650
66	600 E#E&10.	SOLPR660
67	GO TO 800	SOLPR670
68	700 WRITE%6,2< TAU,IQ	SOLPR680
69	GO TO 800	SOLPR690
70	500 WRITE %6,1< TAU,IQ	SOLPR700
71	800 RETURN	SOLPR710
72	END	SOLPR720

TOTAL RECORDS WRITTEN = 73/ 73

WEO LPS